Bitcoin Energy Estimates (DRAFT)

Estimating the energy use of the bitcoin network using two different approaches.

by Steven Black
Project home: https://github.com/StevenBlack/bitcoin-energy-estimates
Updated: October 23 2023

Introduction

Bitcoin mining uses a proof-of-work consensus mechanism. This is controversial for some people because that supposedly requires a lot of electrical energy. We see claims the bitcoin network "*uses as much electricity as a small country*", or "*requires as much electricity as Belgium, or Chile.*"

This Mathematica notebook assessed those notions using the following approaches:

1. ***Presuming bitcoin mining is marginally profitable, how much electical energy can be funded that balances actual mining rewards over time?***
2. ***Given the reported hashrate, how much energy would be required to achieve that?***

This paper uses **Canadian dollars**, partly because that's my fiat currency, and because Canada publishes particularly good statistics about electricity generation and costs.

Background: bitcoin price, block rewards, and fees

Here we outline the major factors that play in the economics of bitcoin mining.

The bitcoin price right now

What is the current price of bitcoin in Canadian dollars?

In[2157]:=

```
Now
```

Out[2157]=

Tue 24 Oct 2023 00:10:18 GMT−4

In[2158]:=

```
BTCPrice =
 CurrencyConvert[Quantity[1, "Bitcoin"], Quantity[1, "CanadianDollars"]]
```

Out[2158]=

C$42 897.58

Bitcoin block rewards

Bitcoin miners are compensated with block rewards from blocks they successfully mine, plus all the transaction fees in that block. In the current epoch (2020 - 2024) the block reward is 6 1/4 BTC.

In[2159]:=
```
blockReward = Quantity[6.25, "BTC"] / Quantity[1, IndependentUnit["block"]]
```
Out[2159]=
```
฿6.25 per block
```

Bitcoin transaction fees per block

ASSUMPTION: the average transaction fees paid to miners, per block, is 0.125 BTC, which is about 2% of the total mining reward for each block.

In[2160]:=
```
blockTransactionFees = Quantity[0.125, "BTC"] / Quantity[IndependentUnit["block"]]
```
Out[2160]=
```
฿0.125 per block
```

Therefore, the total bitcoin paid to miners for an average block, denominated in bitcoin.

In[2161]:=
```
blockRewardPlusFees = (blockReward + blockTransactionFees)
```
Out[2161]=
```
฿6.375 per block
```

The actual block rate

Historically the bitcoin block production rate is faster then the block time target (6 per hour, or 144 blocks per day). Let's reckon an average block rate over a sample interval of blocks in the immediate past.

Let's look at the past 100,000 blocks. How many days did it take to produce the last 100,000 blocks?

In[2162]:=
```
blockchainSampleInterval = 100 000;
sampleIntervalTime =
 Now - BlockchainBlockData[-blockchainSampleInterval]["Timestamp"]
```
Out[2163]=
```
682.269 days
```

Let's calculate the average block time for those 100,000 blocks.

In[2164]:=
```
blockTime = UnitConvert[sampleIntervalTime / blockchainSampleInterval,
  MixedUnit[{"Minutes", "Seconds"}]]
```
Out[2164]=
```
9 min 49.4808 s
```

We can now calculate the actual block rate per hour, and per day

In[2165]:=

```
blockRate =
 Quantity[1, "Hours"] / Quantity[blockTime, "Hours" / IndependentUnit["block"]]
```

Out[2165]=

```
6.10707 block/h
```

In[2166]:=

```
blockRatePerDay = blockRate * Quantity[24, "Hours" / "Day"]
```

Out[2166]=

```
146.57 block/day
```

Total miner compensation, per hour

In[2167]:=

```
blockRewardPlusFeesPerHour = blockRewardPlusFees * blockRate
```

Out[2167]=

```
฿38.9326 per hour
```

Bitcoin price in Canadian dollars as a time series

Let's gather data on bitcoin price over the past sampleIntervalTime.

In[2168]:=

```
btcPriceOverTime = CurrencyConvert["Bitcoin",
   "CanadianDollars", {Now - sampleIntervalTime, Now}];
```

Let's graph that bitcoin price over time.

In[2169]:=

```
DateListPlot[
 btcPriceOverTime,
 PlotLabel → "BTC Price" <> "  — " <> "last " <>
   ToString[blockchainSampleInterval] <> " blocks\nCAD"
 , FrameTicks → {Automatic, All}
 , GridLines → Automatic
 , ImageSize → Large
]
```

Out[2169]=



In[2170]:=

Average bitcoin price over our sample time

In[2171]:=

```
btcPriceAvg = Mean[btcPriceOverTime]
```

Out[2171]=

C$36939.31

1. Assuming mining is ecomomically marginal, how much electricity could be funded by mining rewards?

Global revenue per hour

The value, in Canadian Dollars, of all bitcoin mined globally, per hour.

In[2172]:=
```
blockCADperHour = Quantity[
   QuantityMagnitude[blockRewardPlusFeesPerHour], "per Hour"] * btcPriceAvg
```
Out[2172]=

$C\$1.43814 \times 10^{6}$ per hour

Electricity cost, per kWh

See: https://www.hydroquebec.com/business/customer-space/rates/comparison-electricity-prices.html



Let’s presume that nobody in their right mind would want to mine bitcoin in New York or Boston. Here's the distribution of electricity input costs from the other 5 locations.

In[2173]:=
```
electricityInputCost = Quantity[
   Around[
    {0.0969, 0.1042, 0.1169, 0.1365, 0.1438}
   ]
   , "CanadianDollars"
  ] / Quantity[1, "kWh"]
```
Out[2173]=

C$ (0.120 ± 0.020) per hour per kilowatt

Business cost assumption

Let’s presume 85% of mining revenue is available to pay electricity cost. The rest covers wages, maintenance, depreciation, and taxes.

In[2174]:=

```
availableForElectricity = 0.85
```

Out[2174]=

```
0.85
```

Economically sustainable mining power consumption

In[2175]:=

$$\text{btcPower} = \frac{\text{blockCADperHour} * \text{availableForElectricity}}{\text{electricityInputCost}}$$

Out[2175]=

$$(1.02 \pm 0.17) \times 10^{7}\ \text{kW}$$

Cognitively we can say, bitcoin's power consumption is in the order of 10 GWH.

Economically sustainable mining energy consumption

In[2176]:=

```
AnnualEnergyConsumption = UnitConvert[
  btcPower * Quantity[365 * 24, "Hours"]
  , "TWh"
 ]
```

Out[2176]=

$$(89. \pm 15.)\ \text{h TW}$$

Cognitively we can say, bitcoin's annual energy consumption is between 75 and 105 TWH.

Comparisons with large power generation facilities or regions

Let’s compare the bitcoin network with the power and energy that generated, or used, by various things.

Here's the raw data for various generation facilities and regions.

In[2177]:=

```
generators = {
  <|"name" → "Robert-Bourassa generating station", "capacity" → 5616 MW |>
  , <|"name" → "Grand Coolee dam (USA)", "capacity" → 6809 MW |>
  , <|"name" → "Chile (2021)", "capacity" → 85 h TW / 365 days |>
  , <|"name" → "Belgium (2021)", "capacity" → 95 h TW / 365 days |>
  , <|"name" → "Three Gorges dam (China)", "capacity" → 22 500 MW |>
  , <|"name" → "Province of Québec (2019)", "capacity" → 212.9 h TW / 365 days |>
  , <|"name" → "Canada (2021)", "capacity" → 626 h TW / 365 days |>
  , <|"name" → "USA (2021)", "capacity" → 4165 h TW / 365 days |>
  , <|"name" → "China (2021)", "capacity" → 8152 h TW / 365 days |>
  , <|"name" → "World (2021)", "capacity" → 27 295 h TW / 365 days |>
 };
```

In[2178]:=

```
Grid[
 Prepend[
  Table[
   {#name, btcPower / #capacity} &[generators⟦x⟧]
   , {x, 1, Length[generators]}
  ]
  , {"Facility or region", "BTC relative consumption"}
 ]
 , Frame → All
]
```

Out[2178]=

Facility or region	BTC relative consumption
Robert-Bourassa generating station	(1.82 ± 0.31)
Grand Coolee dam (USA)	(1.50 ± 0.25)
Chile (2021)	(1.05 ± 0.18)
Belgium (2021)	(0.94 ± 0.16)
Three Gorges dam (China)	(0.45 ± 0.08)
Province of Québec (2019)	(0.42 ± 0.07)
Canada (2021)	(0.143 ± 0.024)
USA (2021)	(0.021 ± 0.004)
China (2021)	(0.0110 ± 0.0019)
World (2021)	(0.0033 ± 0.0006)

2. Given the reported hashrate, how much energy would be required to achieve that?

Coming soon.

Conclusion

The claims that the bitcoin network “*uses as much electricity as a small country*”, or “*requires as much electricity as Belgium, or Chile*” seem reasonable from the perspectibe of bitcoin mining economics.

Appendix

Robert-Bourassa generating station — a.k.a. “LG-2”

Here we compare the power consumption of the bitcoin network with the power generation capacity of the Robert Bourassa generating station in the James Bay region of northern Québec.
See https://en.wikipedia.org/wiki/Robert-Bourassa_generating_station

In[2179]:=

```
RobertBourassaDam = 5616 MW // UnitSimplify // N
```

Out[2179]=

```
5.616 GW
```

What is bitcoin’s global energy use in terms of LG-2?

In[2180]:=

```
btcPower / RobertBourassaDam
```

Out[2180]=

```
(1.82 ± 0.31)
```

Province of Québec

In 2019 the Province of Québec produced 212.9 TWh of electricity.

What is bitcoin’s global energy use as a proportion of Québec’s electricity production in 2019?

In[2181]:=

```
Québec2019 = 212.9 h TW
```

Out[2181]=

```
212.9 h TW
```

In[2182]:=

$$\text{Québec2019day} = \frac{\text{Québec2019}}{365\ \text{days}}\ //\ \text{UnitSimplify}$$

Out[2182]=

```
24.3037 GW
```

In[2183]:=

```
btcPower / Québec2019day
```

Out[2183]=

```
(0.42 ± 0.07)
```

Province of Ontario

See https://www.cer-rec.gc.ca/en/data-analysis/energy-markets/provincial-territorial-energy-profiles/provincial-territorial-energy-profiles-ontario.html

In 2019, the average annual power consumption per capita in Ontario was 9.6 megawatt-hours (MWh).

In[2184]:=

```
Ontario2019PerCapita = Quantity[9.6, "Hours" * "Megawatts" / "People"] / Quantity[24 * 365, "Hours"];
Ontario2019PerCapita = UnitConvert[Ontario2019PerCapita, kW / people]
```

Out[2185]=

1.09589 kW/person

In[2186]:=

```
(btcPower / Ontario2019PerCapita)
```

Out[2186]=

$(9.3 \pm 1.6) \times 10^{6}$ people

United States

See https://www.worlddata.info/america/usa/energy-consumption.php

In[2187]:=

```
USAPerCapita = Quantity[11.757, "Hours" * "Megawatts" / "People"] / Quantity[24 * 365, "Hours"];
USAPerCapita = UnitConvert[USAPerCapita, kW / people]
```

Out[2188]=

1.34212 kW/person

In[2189]:=

```
(btcPower / USAPerCapita)
```

Out[2189]=

$(7.6 \pm 1.3) \times 10^{6}$ people

Europe

Again see See https://www.worlddata.info/america/usa/energy-consumption.php

In[2190]:=

```
EuropePerCapita = Quantity[5.462, "Hours" * "Megawatts" / "People"] / Quantity[24 * 365, "Hours"];
EuropePerCapita = UnitConvert[EuropePerCapita, kW / people]
```

Out[2191]=

0.623516 kW/person

In[2192]:=

```
(btcPower / EuropePerCapita)
```

Out[2192]=

$(1.64 \pm 0.28) \times 10^{7}$ people