Bitcoin Energy Estimates (DRAFT)

Estimating the energy use of the Bitcoin network using two different approaches.

by Steven Black
Project home: https://github.com/StevenBlack/bitcoin-energy-estimates
Updated: October 22 2023

Introduction

Bitcoin mining uses a Proof-of-Work consensus mechanism. This is controversial for some because that supposedly requires a lot of electrical energy. We see claims the bitcoin network "*uses as much electricity as a small country*", or *"requires as much electricity as Belgium, or Chile."*

This Mathematica notebook assessed those notions using the following approaches:

1. ***Presuming Bitcoin mining is marginally profitable, how much electical energy can be funded that balances actual mining rewards over time?***
2. ***Given the reported hashrate, how much energy would be required to achieve that?***

This paper uses **Canadian dollars**, partly because that's my fiat currency, and because Canada publishes particularly good statistics about electricity generation and costs.

Bitcoin price, block rewards, and fees

Here we outline the major factors that play in the economics of Bitcoin mining.

The bitcoin price right now

What is the current price of Bitcoin in Canadian dollars?

In[1315]:=

```
Now
```

Out[1315]=

Sun 22 Oct 2023 19:19:35 GMT−4

In[1316]:=

```
BTCPrice =
 CurrencyConvert[Quantity[1, "Bitcoin"], Quantity[1, "CanadianDollars"]]
```

Out[1316]=

C$40 902.16

Bitcoin block rewards

Bitcoin miners are compensated with the block reward from blocks they successfully mine, plus all the transaction fees in that block. In the current epoch (2020 - 2024) the block reward is 6 1/4 BTC.

In[1317]:=
```
blockReward = Quantity[6.25, "BTC"]
```
Out[1317]=
```
฿6.25
```

Bitcoin transaction fees per block

ASSUMPTION: the average of transaction fees per block is 0.15 BTC.

In[1318]:=
```
blockTransactionFees = Quantity[0.15, "BTC"]
```
Out[1318]=
```
฿0.15
```

Therefore, the total Bitcoin paid to miners for an average block, denominated in Bitcoin.

In[1319]:=
```
blockRewardPlusFees = (blockReward + blockTransactionFees)
```
Out[1319]=
```
฿6.4
```

The actual block rate

Historically Bitcoin blocks land at a rate faster then the block time target (6 per hour, or 144 blocks per day). Let's recon an average block rate over a sample interval to present day:

In[1320]:=
```
blockchainSampleInterval = 100 000;
sampleIntervalTime =
 Now - BlockchainBlockData[-blockchainSampleInterval]["Timestamp"]
```
Out[1321]=
```
682.238 days
```

In[1322]:=
```
blockTime = UnitConvert[sampleIntervalTime / blockchainSampleInterval,
  MixedUnit[{"Minutes", "Seconds"}]]
```
Out[1322]=
```
9 min 49.4533 s
```

In[1323]:=
```
blockRate = Quantity[Quantity[1, "Hours"] / blockTime, "per Hour"]
```
Out[1323]=
```
6.10735 per hour
```

Total miner compensation, per hour

In[1324]:=

```
blockRewardPlusFeesPerHour = blockRewardPlusFees * blockRate
```

Out[1324]=

```
฿39.0871 per hour
```

Bitcoin price in Canadian dollars as a time series

Let's gather data on bitcoin price over the past sampleIntervalTime.

In[1325]:=

```
btcPriceOverTime = CurrencyConvert["Bitcoin",
   "CanadianDollars", {Now - sampleIntervalTime, Now}];
```

Let's graph that bitcoin price over time.

In[1326]:=

```
DateListPlot[
 btcPriceOverTime,
 PlotLabel → "BTC Price" <> "  — " <> "last " <>
   ToString[blockchainSampleInterval] <> " blocks\nCAD"
 , FrameTicks → {Automatic, All}
 , GridLines → Automatic
 , ImageSize → Large
]
```

Out[1326]=



In[1327]:=

Average bitcoin price over our sample time

In[1328]:=
```
btcPriceAvg = Mean[btcPriceOverTime]
```
Out[1328]=
```
C$36978.33
```

---

1. Assuming mining is ecomomically marginal, how much electricity could be funded by mining rewards?

Global revenue per hour

The value, in Canadian Dollars, of all Bitcoin mined globally, per hour.

In[1329]:=
```
blockCADperHour = Quantity[
   QuantityMagnitude[blockRewardPlusFeesPerHour], "per Hour"] * btcPriceAvg
```
Out[1329]=

$C\$1.44537\times10^{6}$ per hour

Electricity cost, per kWh

See: https://www.hydroquebec.com/business/customer-space/rates/comparison-electricity-prices.html



Let’s presume that nobody in their right mind would want to mine Bitcoin in New York or Boston. Here's the distribution of electricity input costs from the other 5 locations.

In[1330]:=
```
electricityInputCost = Quantity[
   Around[
    {0.0969, 0.1042, 0.1169, 0.1365, 0.1438}
   ]
   , "CanadianDollars"
  ] / Quantity[1, "kWh"]
```
Out[1330]=
C$ (0.120 ± 0.020) per hour per kilowatt

Business cost assumption

Let’s presume 85% of mining revenue is available to pay electricity cost. The rest covers wages, maintenance, depreciation, and taxes.

In[1331]:=
```
availableForElectricity = 0.85
```
Out[1331]=
0.85

Energy economically sustainable

In[1332]:=
$$\text{btcPower} = \frac{\text{blockCADperHour} * \text{availableForElectricity}}{\text{electricityInputCost}}$$
Out[1332]=
$(1.03 \pm 0.17) \times 10^{7}$ kW

Cognitively we can say, Bitcoin's power consumption is in the order of 10 GWH.

In[1333]:=
```
AnnualEnergyConsumption = UnitConvert[
  btcPower * Quantity[365 * 24, "Hours"]
  , "TWh"
 ]
```
Out[1333]=
(90. ± 15.) h TW

Cognitively we can say, Bitcoin's annual energy consumption is in the order of 90 TWH.

Comparisons with large power generation facilities or regions

Let’s compare the Bitcoin network with the power and energy that generated, or used, by various things.

Here's the raw data for various generation facilities and regions.

In[1334]:=

```
generators = {
   <|"name" → "Robert-Bourassa generating station", "capacity" → 5616 MW |>
  , <|"name" → "Grand Coolee Dam (USA)", "capacity" → 6809 MW |>
  , <|"name" → "Three Gorges Dam (China)", "capacity" → 22 500 MW |>
  , <|"name" → "Province of Québec (2019)", "capacity" → 212.9 h TW / 365 days |>
  };
```

In[1335]:=

```
Table[{#name, btcPower / #capacity} &[generators⟦x⟧], {x, 1, 4}] // Grid // Framed
```

Out[1335]=

Robert-Bourassa generating station	(1.83 ± 0.31)
Grand Coolee Dam (USA)	(1.51 ± 0.25)
Three Gorges Dam (China)	(0.46 ± 0.08)
Province of Québec (2019)	(0.42 ± 0.07)

2. Given the reported hashrate, how much energy would be required to achieve that?

Coming soon.

Appendix

Robert-Bourassa generating station — a.k.a. “LG-2”

Here we compare the power consumption of the bitcoin network with the power generation capacity of the Robert Bourassa generating station in the James Bay region of northern Québec.
See https://en.wikipedia.org/wiki/Robert-Bourassa_generating_station

In[1336]:=

```
RobertBourassaDam = 5616 MW // UnitSimplify // N
```

Out[1336]=

5.616 GW

What is Bitcoin’s global energy use in terms of LG-2?

In[1337]:=

```
btcPower / RobertBourassaDam
```

Out[1337]=

(1.83 ± 0.31)

Province of Québec

In 2019 the Province of Québec produced 212.9 TWh of electricity.

What is Bitcoin's global energy use as a proportion of Québec's electricity production in 2019?

In[1338]:=

```
Québec2019 = 212.9 h TW
```

Out[1338]=

```
212.9 h TW
```

In[1339]:=

```
Québec2019day = Québec2019 / (365 days) // UnitSimplify
```

Out[1339]=

```
24.3037 GW
```

In[1340]:=

```
btcPower / Québec2019day
```

Out[1340]=

```
(0.42 ± 0.07)
```

Province of Ontario

See https://www.cer-rec.gc.ca/en/data-analysis/energy-markets/provincial-territorial-energy-profiles/provincial-territorial-energy-profiles-ontario.html

In 2019, the average annual power consumption per capita in Ontario was 9.6 megawatt-hours (MWh).

In[1341]:=

```
Ontario2019PerCapita = Quantity[9.6, "Hours" * "Megawatts" / "People"] / Quantity[24 * 365, "Hours"];
Ontario2019PerCapita = UnitConvert[Ontario2019PerCapita, kW / people]
```

Out[1342]=

```
1.09589 kW/person
```

In[1343]:=

```
(btcPower / Ontario2019PerCapita)
```

Out[1343]=

$(9.4 \pm 1.6) \times 10^{6}$ people

United States

See https://www.worlddata.info/america/usa/energy-consumption.php

In[1344]:=

```
USAPerCapita = Quantity[11.757, "Hours" * "Megawatts" / "People"] / Quantity[24 * 365, "Hours"];
USAPerCapita = UnitConvert[USAPerCapita, kW / people]
```

Out[1345]=

1.34212 kW/person

In[1346]:=

```
(btcPower / USAPerCapita)
```

Out[1346]=

$(7.6 \pm 1.3) \times 10^{6}$ **people**

Europe

Again see See https://www.worlddata.info/america/usa/energy-consumption.php

In[1347]:=

```
EuropePerCapita = Quantity[5.462, "Hours" * "Megawatts" / "People"] / Quantity[24 * 365, "Hours"];
EuropePerCapita = UnitConvert[EuropePerCapita, kW / people]
```

Out[1348]=

0.623516 kW/person

In[1349]:=

```
(btcPower / EuropePerCapita)
```

Out[1349]=

$(1.65 \pm 0.28) \times 10^{7}$ **people**