Bitcoin Energy Estimates (DRAFT)

Estimating the energy use of the Bitcoin network using various approaches.

by Steven Black
Project home: https://github.com/StevenBlack/bitcoin-energy-estimates
Updated: October 21 2023

Introduction

Bitcoin mining uses a Proof-of-Work consensus mechanism. This is controversial for some because that supposedly requires a lot of electrical energy. We see claims the bitcoin network "*uses as much electricity as a small country*", or *"requires as much electricity as Belgium, or Chile."*

This study assessed those notions using the following approaches:

1. ***Presuming Bitcoin mining is marginally profitable, how much energy can be used compared to actual mining rewards over time?***
2. ***Given the reported hashrate, how much energy would be required to achieve that.***

This paper uses **Canadian dollars**, partly because that's my fiat currency, and because Canada publishes particularly good statistics about electricity generation and costs.

Bitcoin price, block rewards, and fees

Bitcoin price right now

For the purpose of discussion, what is the current price of Bitcoin in Canadian dollars?

In[958]:=
```
Now
```
Out[958]=

Sat 21 Oct 2023 23:41:40 GMT−4

In[959]:=
```
BTCPrice =
 CurrencyConvert[Quantity[1, "Bitcoin"], Quantity[1, "CanadianDollars"]]
```
Out[959]=

C$40 885.41

Bitcoin block rewards

Bitcoin miners are compensated with the block reward for blocks they successfully mine, plus all the transaction fees in that block. In the current epoch (2020 - 2024) the block reward is 6 1/4 BTC.

In[960]:=
```
blockreward = Quantity[6.25, "BTC"]
```
Out[960]=
```
฿6.25
```

ASSUMPTION: the average of transaction fees per block is 0.15 BTC.

In[961]:=
```
blockfees = Quantity[0.15, "BTC"]
```
Out[961]=
```
฿0.15
```

Therefore, the total Bitcoin paid to miners for an average block, denominated in Bitcoin.

In[962]:=
```
blockRewardPlusFees = (blockreward + blockfees)
```
Out[962]=
```
฿6.4
```

The actual block rate

Historically Bitcoin blocks land at a rate faster then the block time target (6 per hour, or 144 blocks per day). Let's recon an average block rate over a sample interval to present day:

In[963]:=
```
blocksample = 100 000;
sampletime = Now - BlockchainBlockData[-blocksample]["Timestamp"]
```
Out[964]=
```
682.141 days
```

In[965]:=
```
blocktime =
 UnitConvert[sampletime / blocksample, MixedUnit[{"Minutes", "Seconds"}]]
```
Out[965]=
```
9 min 49.3694 s
```

In[966]:=
```
blockrate = Quantity[Quantity[1, "Hours"] / blocktime, "per Hour"]
```
Out[966]=
```
6.10822 per hour
```

In[967]:=
```
blockRewardPlusFeesPerHour = blockRewardPlusFees * blockrate
```
Out[967]=
```
฿39.0926 per hour
```

Bitcoin price as a time series

Let's gather data on bitcoin price over the past sampletime.

In[968]:=
```
btcpriceovertime =
  CurrencyConvert["Bitcoin", "CanadianDollars", {Now - sampletime, Now}];
```

In[969]:=
```
btcpriceavg = Mean[btcpriceovertime]
```

Out[969]=
```
C$37 035.15
```

Hourly Economics

Global revenue per hour

The value, in Canadian Dollars, of all Bitcoin mined globally, per hour.

In[970]:=
```
blockCADperHour = Quantity[
   QuantityMagnitude[blockRewardPlusFeesPerHour], "per Hour"] * btcpriceavg
```

Out[970]=
```
C$1.4478 × 10^6 per hour
```

Electricity cost, per kWh

See: https://www.hydroquebec.com/business/customer-space/rates/comparison-electricity-prices.html

The figures below show a comparison of electricity average prices for four consumption levels in major Nort American cities.

Average prices for electricity (¢/kWh)

Consumption: 10,000 kWh/month
Power demand: 40 kW



Let's presume that nobody in their right mind would want to mine Bitcoin in New York or Boston. Here's the distribution of electricity input costs from the other 5 locations.

In[971]:=
```
electricityInputCost = Quantity[
   Around[
    {0.0969, 0.1042, 0.1169, 0.1365, 0.1438}
   ]
   , "CanadianDollars"
  ] / Quantity[1, "kWh"]
```

Out[971]=
$$\text{C\$}\,(0.120 \pm 0.020)\ \text{per hour per kilowatt}$$

Business cost assumption

Let’s presume 85% of mining revenue is available to pay electricity cost.

In[972]:=
```
availableForElectricity = 0.85
```

Out[972]=
$$0.85$$

Energy economically sustainable

In[973]:=
$$\text{btcPower} = \frac{\text{blockCADperHour} * \text{availableForElectricity}}{\text{electricityInputCost}}$$

Out[973]=
$$(1.03 \pm 0.17) \times 10^{7}\ \text{kW}$$

Cognitively we can say, Bitcoin's power consumption is in the order of 11 GWH.

In[974]:=
```
AnnualEnergyConsumption = btcPower * Quantity[365 * 24, "Hours"]
```

Out[974]=
$$(9.0 \pm 1.5) \times 10^{10}\ \text{h kW}$$

Comparisons with large power generation facilities or regions

Let’s compare the Bitcoin network with the power and energy that generated, or used, by various things.

Here's the raw data for various generation facilities and regions.

In[975]:=
```
generators = {
   <|"name" → "Robert-Bourassa generating station", "capacity" → 5616 MW |>
   , <|"name" → "Grand Coolee Dam (USA)", "capacity" → 6809 MW |>
   , <|"name" → "Three Gorges Dam (China)", "capacity" → 22 500 MW |>
   , <|"name" → "Province of Québec (2019)", "capacity" → 212.9 h TW / 365 days |>
  };
```

In[976]:=
```
Table[{#name, btcPower / #capacity} &[generators[[x]]], {x, 1, 4}] // Grid // Framed
```

Out[976]=

Robert-Bourassa generating station	(1.83 ± 0.31)
Grand Coolee Dam (USA)	(1.51 ± 0.25)
Three Gorges Dam (China)	(0.46 ± 0.08)
Province of Québec (2019)	(0.42 ± 0.07)

Robert-Bourassa generating station — a.k.a. “LG-2”

Here we compare the power consumption of the bitcoin network with the power generation capacity of the Robert Bourassa generating station in the James Bay region of northern Québec.
See https://en.wikipedia.org/wiki/Robert-Bourassa_generating_station

In[977]:=
```
RobertBourassaDam = 5616 MW // UnitSimplify // N
```

Out[977]=
```
5.616 GW
```

What is Bitcoin’s global energy use in terms of LG-2?

In[978]:=
```
btcPower / RobertBourassaDam
```

Out[978]=
```
(1.83 ± 0.31)
```

Province of Québec

In 2019 the Province of Québec produced 212.9 TWh of electricity.

What is Bitcoin’s global energy use as a proportion of Québec’s electricity production in 2019?

In[979]:=
```
Québec2019 = 212.9 h TW
```

Out[979]=
```
212.9 h TW
```

In[980]:=
```
Québec2019day = Québec2019 / 365 days // UnitSimplify
```
Out[980]=
```
24.3037 GW
```

In[981]:=
```
btcPower / Québec2019day
```
Out[981]=
```
(0.42 ± 0.07)
```

Province of Ontario

See https://www.cer-rec.gc.ca/en/data-analysis/energy-markets/provincial-territorial-energy-profiles/provincial-territorial-energy-profiles-ontario.html

In 2019, the average annual power consumption per capita in Ontario was 9.6 megawatt-hours (MWh).

In[982]:=
```
Ontario2019PerCapita = Quantity[9.6, "Hours" * "Megawatts" / "People"] / Quantity[24 * 365, "Hours"];
Ontario2019PerCapita = UnitConvert[Ontario2019PerCapita, kW / people]
```
Out[983]=
```
1.09589 kW/person
```

In[984]:=
```
(btcPower / Ontario2019PerCapita)
```
Out[984]=
```
(9.4 ± 1.6) × 10^6 people
```

United States

See https://www.worlddata.info/america/usa/energy-consumption.php

In[985]:=
```
USAPerCapita = Quantity[11.757, "Hours" * "Megawatts" / "People"] / Quantity[24 * 365, "Hours"];
USAPerCapita = UnitConvert[USAPerCapita, kW / people]
```
Out[986]=
```
1.34212 kW/person
```

In[987]:=
```
(btcPower / USAPerCapita)
```
Out[987]=
```
(7.7 ± 1.3) × 10^6 people
```

Europe

Again see See https://www.worlddata.info/america/usa/energy-consumption.php

In[988]:=

```
EuropePerCapita = Quantity[5.462, "Hours" * "Megawatts" / "People"] / Quantity[24 * 365, "Hours"];
EuropePerCapita = UnitConvert[EuropePerCapita, kW / people]
```

Out[989]=

$$0.623516\ \text{kW/person}$$

In[990]:=

```
(btcPower / EuropePerCapita)
```

Out[990]=

$$(1.65 \pm 0.28) \times 10^7\ \text{people}$$